
[bookmark: _Toc122434485][bookmark: _Toc93054710]Overview
This document lists all the essential changes needed to correct problems in the TTCN implementation of test case 5.4.MCPTT which is part of the ATS iwd-TTCN3-B2020-09_D21wk49 test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 
Contact:	Fidel Liberal
	fidel.liberal@ehu.eus
[bookmark: _Toc84500498][bookmark: _Toc93054711]Verification Test Summary 
Test Case: 	5.4.MCPTT
ATS Version:	ATS iwd-TTCN3-B2020-09_D21wk49
System Simulator used:	MCSTaaSting Tester
UE used:	Nemergent Client v7.0 on COTS LTE UE
Verification Status:	PASS

[bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc54888062][bookmark: _Toc77086936][bookmark: _Toc84500499][bookmark: _Toc93054712][bookmark: _Toc68002782]Corrections required
Following changes are related to the need of a global entry in MCX_Global_Type to hold Floor Control mc_priority value.  This is needed at step 7 to 9 of TC 5.4:  A new mc-priority is set via UPDATE from the SS at step 7 which shall match the value in the 200 OK from the UE at step 8. Also at step 9 again a new value is used in the INVITE from the SS which shall match the mc-priority in the following 200 OK from the UE.
[bookmark: __RefHeading___Toc129176_2187938929][bookmark: _Toc93054713]Change 1
	Record name
	MCX_Global_Type

	Reason for change
	Need of a global entry in MCX_Global_Type to hold Floor Control mc_priority value.  This is needed at step 7 to 9 of TC 5.4:  A new mc-priority is set via UPDATE from the SS at step 7 which shall match the value in the 200 OK from the UE at step 8. Also at step 9 again a new value is used in the INVITE from the SS which shall match the mc-priority in the following 200 OK from the UE

	Summary of change
	New entry added in record MCX_Global_Type

	TTCN module
	MCX_IMS_Component.ttcn

	MCC160 Comment
	Not needed



[bookmark: OLE_LINK1412412111225]Before Change:
	 type record MCX_Global_Type {                                                 /* @status    APPROVED (MCX) */
    MCX_SecurityParameters_Type SecurityParameters optional,
    MCX_GroupInfoList_Type      GroupInfoList,
    charstring                  ClientId optional,
    PortNumber_Type             MediaControlPortUE,
    integer                     FloorControlSeqNums[2],
    O4_Type                     SSRC_UE optional,
  };


[bookmark: OLE_LINK14124121112251]
After Change:
	 type record MCX_Global_Type {                                                 /* @status    APPROVED (MCX) */
    MCX_SecurityParameters_Type SecurityParameters optional,
    MCX_GroupInfoList_Type      GroupInfoList,
    charstring                  ClientId optional,
    PortNumber_Type             MediaControlPortUE,
    integer                     FloorControlSeqNums[2],
    O4_Type                     SSRC_UE optional,
    integer                     FloorControl_mc_priority optional
  };


[bookmark: __RefHeading___Toc129178_2187938929][bookmark: _Toc93054714]Change 2
	Function name
	 fl_MCX_SDP_MediaDescription_CheckFloorControl

	Reason for change
	See Change 1

	Summary of change
	Mc-priority of v_ExpectedFloorControlParameters depends on the value set in global record vc_Global.FloorControl_mc_priority

	TTCN module
	MCX_SDP_Messages.ttcn

	MCC160 Comment
	Accepted in principle



[bookmark: OLE_LINK141241211122511]Before Change:
	fl_MCX_SDP_MediaDescription_CheckFloorControl (…)
{
…
 
   // Check floor control parameters
    select (p_OfferAnswer) {
      case (SDP_OFFER) {
        if (p_IsInitialOffer) {
          v_InitialOffer := true;
        }
        v_ExpectedFloorControlParameters := cr_MCPTT_SDP_FloorControlParameters_OFFER(v_Queuing, v_InitialOffer, p_ImplicitGrantRequest);
      }
      case (SDP_ANSWER) {
	 v_ExpectedFloorControlParameters := cr_MCPTT_SDP_FloorControlParameters_ANSWER(v_Queuing);
      }
    }
 
 ...


[bookmark: OLE_LINK141241211122512]
After Change:
	fl_MCX_SDP_MediaDescription_CheckFloorControl (…)
{
…
 
   // Check floor control parameters
    select (p_OfferAnswer) 
      case (SDP_OFFER) {
        if (p_IsInitialOffer) {
          v_InitialOffer := true;
        }
        v_ExpectedFloorControlParameters := cr_MCPTT_SDP_FloorControlParameters_OFFER(v_Queuing, v_InitialOffer, p_ImplicitGrantRequest);
      }
      case (SDP_ANSWER) {
	 if (isvalue(vc_Global.FloorControl_mc_priority)) { 
	 v_ExpectedFloorControlParameters := cr_MCPTT_SDP_FloorControlParameters_ANSWER(v_Queuing, vc_Global.FloorControl_mc_priority);
	 } else {
	 v_ExpectedFloorControlParameters := cr_MCPTT_SDP_FloorControlParameters_ANSWER(v_Queuing);
	 }
      }
    }



[bookmark: __RefHeading___Toc129180_2187938929][bookmark: _Toc93054715]Change 3
	Template name
	 cs_MCX_IMS_PTC_Global_Init

	Reason for change
	See Change 1 

	Summary of change
	Entry FloorControl_mc_priority set to omit in initialization template

	TTCN module
	MCX_IMS_Component.ttcn

	MCC160 Comment
	Not needed



[bookmark: OLE_LINK14124121112252]Before Change:
	  template (value) MCX_Global_Type cs_MCX_IMS_PTC_Global_Init :=
  { /* @status    APPROVED (MCX) */
    SecurityParameters := omit,
    GroupInfoList := {},
    ClientId := omit,
    MediaControlPortUE := 0,
    FloorControlSeqNums := { 0, 0 },
    SSRC_UE := omit
    
  };


[bookmark: OLE_LINK141241211122521]
After Change:
	  template (value) MCX_Global_Type cs_MCX_IMS_PTC_Global_Init :=
  { /* @status    APPROVED (MCX) */
    SecurityParameters := omit,
    GroupInfoList := {},
    ClientId := omit,
    MediaControlPortUE := 0,
    FloorControlSeqNums := { 0, 0 },
    SSRC_UE := omit,
    FloorControl_mc_priority := omit 
    
  };


[bookmark: __RefHeading___Toc129180_21879389291][bookmark: _Toc93054716]Change 4
	Function name
	 f_TC_5_4_MCPTT_MCX_IMS

	Reason for change
	See Change 1

	Summary of change
	 - Mc-priority value is set in global record in order to use the correct value when matching SDP/FloorControl received contents.
- ‘GROUP_CALL’ must be passed as parameter to function f_MCX_InviteMT_SendINVITE


	TTCN module
	MCPTT_ClientConfiguration_MCX_IMS.ttcn

	MCC160 Comment
	- Accepted in principle
- GROUP_CALL shall not be used as there is no group call in this test case; furthermore RTC/RTCP for audio shall not be configured in case of preestablished session as long as there is no call (yet) 
 see MCC160 Implementation for configuration of SRTP below.



[bookmark: OLE_LINK1412412111225211]Before Change:
	 function f_TC_5_4_MCPTT_MCX_IMS(...)
{
…
    // @siclog "Steps 7 - 8" siclog@
    v_MC_Priority := 2;
    v_SdpMessageTX := f_MCPTT_SdpOffer_TX(-, v_FmtAudio, -, v_MC_Priority);
    v_SdpInfoRX := f_MCPTT_SdpInfo_RX(-, SDP_ANSWER);
    v_ContactSS := f_MCX_CallSetupMO_ContactSS(MCPTT, v_SessionID);          // same Contact as in the response for the INVITE creating the dialog (200 OK)
    f_MCX_SendUPDATE_Receive200OK(v_InviteRequestWithSdpMO, v_ContactSS, v_SdpMessageTX, v_SdpInfoRX);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 8");

    v_MC_Priority := 3;
    v_SdpMessageTX := f_MCPTT_SdpOffer_TX(-, v_FmtAudio, -, v_MC_Priority);
    // @siclog "Step 9" siclog@
    v_InviteRequestWithSdpMT := f_MCX_InviteMT_SendINVITE(MCPTT, -, -, -, v_ContactSS, -, v_SdpMessageTX, -, tsc_MCX_ReInvite);
    f_MCPTT_InviteMT_Receive200OK_SendACK(-, v_InviteRequestWithSdpMT, -, PRE_EST_SESSION_ESTAB_MOD, v_SdpInfoRX);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 9");

    // @siclog "Step 10 Void" siclog@
...
  }


[bookmark: OLE_LINK1412412111225212]
After Change:
	function f_TC_5_4_MCPTT_MCX_IMS(...)
{
…
    // @siclog "Steps 7 - 8" siclog@
    v_MC_Priority := 2;
      vc_Global.FloorControl_mc_priority := v_MC_Priority;
    v_SdpMessageTX := f_MCPTT_SdpOffer_TX(-, v_FmtAudio, -, v_MC_Priority);
    v_SdpInfoRX := f_MCPTT_SdpInfo_RX(-, SDP_ANSWER);
    v_ContactSS := f_MCX_CallSetupMO_ContactSS(MCPTT, v_SessionID);          // same Contact as in the response for the INVITE creating the dialog (200 OK)
    f_MCX_SendUPDATE_Receive200OK(v_InviteRequestWithSdpMO, v_ContactSS, v_SdpMessageTX, v_SdpInfoRX);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 8");

    v_MC_Priority := 3;
     vc_Global.FloorControl_mc_priority := v_MC_Priority;
    v_SdpMessageTX := f_MCPTT_SdpOffer_TX(-, v_FmtAudio, -, v_MC_Priority);
    // @siclog "Step 9" siclog@
    v_InviteRequestWithSdpMT := f_MCX_InviteMT_SendINVITE(MCPTT, GROUP_CALL, -, -, v_ContactSS, -, v_SdpMessageTX, -, tsc_MCX_ReInvite);
    f_MCPTT_InviteMT_Receive200OK_SendACK(-, v_InviteRequestWithSdpMT, -, PRE_EST_SESSION_ESTAB_MOD, v_SdpInfoRX);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 9");

    // @siclog "Step 10 Void" siclog@
...
  }



[bookmark: _Toc93054717]MCC160 Implementation for mc_priority
	  type record MCX_SdpCheckInfo_Type {                                                   /* @status    APPROVED (MCX) */
    MCPTT_SDP_OfferAnswer_Type          OfferAnswer,
    SDP_attribute                       Direction,
    MCX_MediaControlCondition_Type      FloorControlCondition,
    boolean                             EndToEndSecurity,
    integer                             NonDefaultPriority optional
  };

	  template (value) MCX_SdpCheckInfo_Type cs_MCX_SdpCheckInfo(MCPTT_SDP_OfferAnswer_Type p_OfferAnswer,
                                                             template (value) SDP_attribute p_Direction,
                                                             MCX_MediaControlCondition_Type p_FloorControlCondition,
                                                             boolean p_EndToEndSecurity) :=
  { /* @status    APPROVED (MCX) */
    OfferAnswer            := p_OfferAnswer,
    Direction              := p_Direction,
    FloorControlCondition  := p_FloorControlCondition,
    EndToEndSecurity       := p_EndToEndSecurity,
    NonDefaultPriority     := omit
  };

	  template (present) MCPTT_SDP_FloorControlParameters_Type cr_MCPTT_SDP_FloorControlParameters_OFFER(template Null_Type p_Queuing,
                                                                                                     template integer p_Priority,
                                                                                                     template Null_Type p_InitialOffer,
                                                                                                     template Null_Type p_ImplicitRequest) :=
  { /* @status    APPROVED (MCX) */
    mc_queueing := p_Queuing,
    mc_priority := p_Priority(1..255),
    mc_granted := p_InitialOffer,
    mc_implicit_request := p_ImplicitRequest,
    mc_ssrc := omit
  };

	  template (present) MCPTT_SDP_FloorControlParameters_Type cr_MCPTT_SDP_FloorControlParameters_ANSWER(template Null_Type p_Queuing,
                                                                                                      template integer p_Priority := tsc_MCX_User_A_Priority) :=
  { /* @status    APPROVED (MCX) */
    mc_queueing := p_Queuing,
    mc_priority := p_Priority,
    mc_granted := omit,
    mc_implicit_request := omit,
    mc_ssrc := omit
  };

	  function fl_MCX_SDP_MediaDescription_CheckFloorControl(SDP_media_desc p_MediaDescription,
                                                         MCPTT_SDP_OfferAnswer_Type p_OfferAnswer,
                                                         boolean p_IsInitialOffer,
                                                         template Null_Type p_ImplicitGrantRequest,
                                                         boolean p_PRE_ESTABLISHED_SESSION,
                                                         charstring p_IpAddr,
                                                         template (omit) integer p_NonDefaultPriority := omit) runs on IMS_PTC
  {
    var MCPTT_SDP_FloorControlParameters_Type v_FloorControlParameters := f_MCX_SDP_MediaDescription_FloorControlDecodeParameters(p_MediaDescription);
    var template (present) SDP_attribute_list v_AllowedMediaAttributes := { cr_SDP_Attribute_fmtp(tsc_MCPTT_FmtFloorControl) };
    var template (present) MCPTT_SDP_FloorControlParameters_Type v_ExpectedFloorControlParameters;
    var template (omit) Null_Type v_InitialOffer := omit;
    var template (omit) Null_Type v_Queuing := omit;
    var template integer v_Priority;

    if (pc_MCPTT_FloorRequestQueueing) {   // @sic R5-217965 sic@
      v_Queuing := true;
    }
    if (ispresent(p_NonDefaultPriority)) {
      v_Priority := p_NonDefaultPriority;
    }

    // Check floor control parameters
    select (p_OfferAnswer) {
      case (SDP_OFFER) {
        if (p_IsInitialOffer) {
          v_InitialOffer := true;
        }
        if (not isbound(v_Priority)) { v_Priority := (1..255); }                  // any allowed value
        v_ExpectedFloorControlParameters := cr_MCPTT_SDP_FloorControlParameters_OFFER(v_Queuing, v_Priority, v_InitialOffer, p_ImplicitGrantRequest);
      }
      case (SDP_ANSWER) {
        if (not isbound(v_Priority)) { v_Priority := tsc_MCX_User_A_Priority; }  // same value as used in offer per default
        v_ExpectedFloorControlParameters := cr_MCPTT_SDP_FloorControlParameters_ANSWER(v_Queuing, v_Priority);
      }
    }
    if (not match(v_FloorControlParameters, v_ExpectedFloorControlParameters)) {
      f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "invalid floor control parameters");
    }

    // Check that there are no unappropriate attributes
    v_AllowedMediaAttributes := f_SDP_Attributes_Add_RX(v_AllowedMediaAttributes, cr_SDP_Attribute_candidate_any, p_PRE_ESTABLISHED_SESSION);
    fl_MCX_SDP_MediaDescription_CheckAllowedAttributes(p_MediaDescription, v_AllowedMediaAttributes);

    // Check ICE candidates
    if (p_PRE_ESTABLISHED_SESSION) {
      fl_MCX_SDP_CheckIceCandidates(p_MediaDescription, 1, p_IpAddr, p_MediaDescription.media_field.ports.port_number);
    }
  }

	  function f_MCPTT_CheckSDP(template (omit) SDP_Message p_SdpMessage,
                            template MCX_SdpInfoRX_Type p_SdpInfoRX) runs on MCX_IMS_PTC
  { /* p_SdpInfoRX  ..  omit if there shall be no SDP message */
    var boolean v_PRE_ESTABLISHED_SESSION := f_GetTestcaseAttrib_PRE_ESTABLISHED_SESSION(testcasename());
    var boolean v_IsInitialOffer := not isvalue(f_IMS_PTC_ImsInfo_DialogGetPrevSdpMessageRX()); /* NOTE: f_IMS_PTC_ImsInfo_DialogGetPrevSdpMessageRX needs to be called before f_IMS_SIP_DecodeMatchAndCheckSDP,
                                                                                                   as f_IMS_SIP_DecodeMatchAndCheckSDP will set SdpMessagePrevRX */
    var charstring v_IPAddrUE := f_IP_AddrInfo_AddrStr(f_IMS_PTC_UE_Address_Get());             /* IP address of the UE: used as unicast-address in the origin, as session-level connection-address (if present) and as media-level connection-address (if present) */
    var MCX_SdpCheckInfo_Type v_CheckInfo;
    var SDP_Message v_SDP_Message;
    var template (omit) SDP_media_desc v_MediaDescAudio;
    var template (omit) SDP_media_desc v_MediaDescApplication;
    var template Null_Type v_ImplicitGrantRequest;
    var SDP_media_desc v_MediaDescValue;
    var PortNumber_Type v_PortNumberUE;
    var MCX_SecurityKey_Type v_PCK;
    
    if (ispresent(p_SdpInfoRX.CheckInfo) and isvalue(p_SdpMessage)) {
      v_SDP_Message := valueof(p_SdpMessage);
      v_CheckInfo := valueof(p_SdpInfoRX.CheckInfo);
      v_MediaDescAudio := f_MCX_SDP_GetAudioDescription(v_SDP_Message);
      v_MediaDescApplication := f_MCX_SDP_GetMediaControlDescription(v_SDP_Message);
      
      // Check Connections
      f_MCX_SDP_CheckIpAddresses(v_SDP_Message, v_IPAddrUE);     /* check IP addresses in origin and c= lines at session and media level;
                                                                    NOTE: f_IMS_SDP_CheckConnections checks that there is a session-level address or a media-level address for each media or both, but it does not check the value */
      if (isvalue(v_MediaDescAudio)) {
        v_MediaDescValue := valueof(v_MediaDescAudio);
        fl_MCX_SDP_MediaDescription_CheckAudio(v_MediaDescValue, v_PRE_ESTABLISHED_SESSION, v_IPAddrUE, v_CheckInfo.Direction);
      }
      if (isvalue(v_MediaDescApplication)) {
        select (v_CheckInfo.FloorControlCondition) {
          case (floorControlWithoutImplictGrantRequest) { v_ImplicitGrantRequest := omit; }
          case (floorControlWithImplictGrantRequest)    { v_ImplicitGrantRequest := true; }
          case (withOptionalMediaControl)               { v_ImplicitGrantRequest := *; }
          case else { f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "unexpected media description for floor control"); }
        }
        v_MediaDescValue := valueof(v_MediaDescApplication);
        fl_MCX_SDP_MediaDescription_CheckFloorControl(v_MediaDescValue, v_CheckInfo.OfferAnswer, v_IsInitialOffer, v_ImplicitGrantRequest, v_PRE_ESTABLISHED_SESSION, v_IPAddrUE, v_CheckInfo.NonDefaultPriority);
        v_PortNumberUE := v_MediaDescValue.media_field.ports.port_number;
        f_MCX_IMS_PTC_UE_MediaControlPort_Set(v_PortNumberUE);
      }
      if (v_CheckInfo.EndToEndSecurity) {
        v_PCK := fl_MCPTT_SDP_GetPCK(v_SDP_Message);
        f_MCX_IMS_PTC_SecurityParameters_SetKey(PCK, v_PCK);
      }
    }
  }

	  function f_TC_5_4_MCPTT_MCX_IMS() runs on MCX_IMS_PTC
  { // Configuration / Pre-established Session Establishment / Pre-established Session Modification / Pre-established Session Release
…
    // @siclog "Steps 7 - 8" siclog@
    v_MC_Priority := 2;
    v_SdpMessageTX := f_MCPTT_SdpOffer_TX(-, v_FmtAudio, -, v_MC_Priority);
    v_SdpInfoRX := f_MCPTT_SdpInfo_RX(-, SDP_ANSWER);
    v_SdpInfoRX.CheckInfo.NonDefaultPriority := v_MC_Priority;
    v_ContactSS := f_MCX_CallSetupMO_ContactSS(MCPTT, v_SessionID);          // same Contact as in the response for the INVITE creating the dialog (200 OK)
    f_MCPTT_SendUPDATE_Receive200OK(v_InviteRequestWithSdpMO, v_ContactSS, v_SdpMessageTX, v_SdpInfoRX);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 8");

    v_MC_Priority := 3;
    v_SdpMessageTX := f_MCPTT_SdpOffer_TX(-, v_FmtAudio, -, v_MC_Priority);
    v_SdpInfoRX.CheckInfo.NonDefaultPriority := v_MC_Priority;

    // @siclog "Step 9" siclog@
    v_InviteRequestWithSdpMT := f_MCX_InviteMT_SendINVITE(MCPTT, -, -, -, v_ContactSS, -, v_SdpMessageTX, -, tsc_MCX_ReInvite);
    f_MCPTT_InviteMT_Receive200OK_SendACK(-, v_InviteRequestWithSdpMT, -, PRE_EST_SESSION_ESTAB_MOD, v_SdpInfoRX);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 9");
…
  }



MCC160 Implementation for configuration of SRTP
	  function f_MCX_ConfigSRTP(template (omit) MCX_CS_ID_Type p_CS_Id_RTP) runs on MCX_IMS_PTC
  {
    var IP_AddrInfo_Type v_IP_AddrInfo := f_IMS_PTC_NW_Address_Get();
    var O4_Type v_LocalSSRC := tsc_MCX_SSRC_NW;
    var MCX_CS_ID_Type v_CS_Id_RTCP := CSK_MCPTT;
    var MCX_KeyIndex_Type v_KeyIndex_RTCP := CSK;
    var template (value) SRTP_SecurityProfile_Type v_SecurityProfile_RTCP := cds_SRTP_SecurityProfile_CSK;
    var template (omit) SRTP_Config_Type v_SRTP_Config_RTCP := f_MIKEY_KeyDerivation(v_KeyIndex_RTCP, v_CS_Id_RTCP, v_SecurityProfile_RTCP);

    var MCX_CS_ID_Type v_CS_Id_RTP;
    var MCX_KeyIndex_Type v_KeyIndex_RTP;
    var template (value) SRTP_SecurityProfile_Type v_SecurityProfile_RTP;
    var template (omit) SRTP_Config_Type v_SRTP_Config_RTP;

    var template (value) SRTP_ConnectionTable_Type v_SRTP_ConnectionTable;

    if (isvalue(p_CS_Id_RTP)) {
      v_CS_Id_RTP := valueof(p_CS_Id_RTP);

      select (pv_CS_Id_RTP) {
        case (PCK_MCPTT_Initiator, PCK_MCPTT_Receiver) {
          v_KeyIndex_RTP := PCK;
          v_SecurityProfile_RTP := cds_SRTP_SecurityProfile_PCK;
        }
        case (GMK_MCPTT) {
          v_KeyIndex_RTP := GMK;
          v_SecurityProfile_RTP := cds_SRTP_SecurityProfile_GMK;
        }
        case else {
          FatalError(__FILE__, __LINE__, "invalid CS_ID");
        }
      }
      v_SRTP_Config_RTP := f_MIKEY_KeyDerivation(v_KeyIndex_RTP, pv_CS_Id_RTP, v_SecurityProfile_RTP);
    v_SRTP_Config_RTCP := f_MIKEY_KeyDerivation(v_KeyIndex_RTCP, v_CS_Id_RTCP, v_SecurityProfile_RTCP);

      v_SRTP_ConnectionTable := {
        cs_SRTP_Connection(cs_UDP_ConnectionId(cs_IP_Socket(v_IP_AddrInfo, tsc_MCX_MediaPort_Audio)),     v_LocalSSRC, v_SRTP_Config_RTP),
        cs_SRTP_Connection(cs_UDP_ConnectionId(cs_IP_Socket(v_IP_AddrInfo, tsc_MCX_MediaPort_Audio + 1)), v_LocalSSRC, v_SRTP_Config_RTCP),
        cs_SRTP_Connection(cs_UDP_ConnectionId(cs_IP_Socket(v_IP_AddrInfo, tsc_MCX_MediaControlPortNW)),  v_LocalSSRC, v_SRTP_Config_RTCP)
      };
    }

    v_SRTP_ConnectionTable[lengthof(v_SRTP_ConnectionTable)] := cs_SRTP_Connection(cs_UDP_ConnectionId(cs_IP_Socket(v_IP_AddrInfo, tsc_MCX_MediaControlPortNW)), v_LocalSSRC, v_SRTP_Config_RTCP);

    SRTP.send(cas_SRTP_CTRL_REQ(v_SRTP_ConnectionTable));
    alt {
      [] SRTP.receive(car_SRTP_CTRL_IND_Confirm) {}
      [] SRTP.receive(car_SRTP_CTRL_IND_Any) {
        f_SetVerdictInconc(__FILE__, __LINE__, "error configuring SRTP");
      }
    }
  }

	  function f_MCX_CallSetup_ConfigSRTP(MCX_CallType_Type p_CallType,
                                      IMS_TestProcedure_Type p_TestProcedure) runs on MCX_IMS_PTC
  {
    var template (omit) MCX_CS_ID_Type v_CS_Id_RTP := omit;

    select (p_CallType) {
      case (GROUP_CALL, CHAT_GROUP_CALL) {
        v_CS_Id_RTP := GMK_MCPTT;
      }
      case (PRIVATE_CALL) {
        select (p_TestProcedure) {
          case (IPCAN_MT_SpeechCall) { v_CS_Id_RTP := PCK_MCPTT_Receiver; }  // @sic R5s210398 - Additional change: PCK_MCPTT_Receiver instead of PCK_MCPTT_Initiator sic@
          case (IPCAN_MO_SpeechCall) { v_CS_Id_RTP := PCK_MCPTT_Initiator; }
          case else { FatalError(__FILE__, __LINE__, ""); }
        }
      }
      case (FIRST_TO_ANSWER) {
        select (p_TestProcedure) {
          case (IPCAN_MT_SpeechCall) { v_CS_Id_RTP := PCK_MCPTT_Initiator; }
          case (IPCAN_MO_SpeechCall) { v_CS_Id_RTP := PCK_MCPTT_Receiver; }
          case else { FatalError(__FILE__, __LINE__, ""); }
        }
      }
    }
    f_MCX_ConfigSRTP(v_CS_Id_RTP);   // shall not be done before PCK is stored
  }



Additional change
In the optional branch of steps 4a1-5b1 according to Table 5.4.3.3-5B the SDP offer sent by the UE shall be according to condition IMPLICIT_GRANT_REQUESTED what is missingin TTCN so far.
	  function f_TC_5_4_MCPTT_MCX_IMS() runs on MCX_IMS_PTC
  { // Configuration / Pre-established Session Establishment / Pre-established Session Modification / Pre-established Session Release
…
    // @siclog "Step 4" siclog@
    if (pc_MCX_UserInitiatedPreestablishedSessionModification) {
    
      // @siclog "Step 4a2" siclog@
      f_MCX_IPCAN_StartSignalling(IPCAN_MO_IMS_Signalling);                     // @sic R5-213659 sic@

      // @siclog "Step 4a1" siclog@
      f_UT_MCX_UserAction(MMI, "Make the MCPTT User request to modify the pre-established session by requesting usage of implicit floor control");
      
      v_SdpInfoRX := f_MCPTT_SdpInfo_RX(-, SDP_OFFER, floorControlWithImplictGrantRequest);
      
      alt {
        [] IMS_Server.check(receive(car_IMS_Update_Request(cr_UPDATE_Request(?)))) {
          // @siclog "Steps 5a1 - 5a2" siclog@
          f_MCPTT_ReceiveUPDATE_Send200OK(v_InviteRequestWithSdpMO, v_SessionID, v_SdpInfoRX);
          f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 5a1");
        }
        [] IMS_Server.check(receive(car_IMS_Invite_Request(cr_INVITE_Request(?)))) {
          // @siclog "Steps 5b" siclog@
          f_MCPTT_PreestablishedSession_Establishment(-, v_SessionID, v_SessionID, v_SdpInfoRX, tsc_MCX_ReInvite);
          f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 5b1");
        }
      }
    } else {
      // @siclog "Step 6Aa1" siclog@
      f_MCX_IPCAN_StartSignalling(IPCAN_MT_IMS_Signalling);                     // @sic R5-213659 sic@
    }
…
  }





Branches executed 
· TLSv1.2
· IPSec ipsec-3gpp;alg=hmac-sha-1-96;ealg=aes-cbc
· XML integrity and encrpytion according to  http://www.w3.org/2009/xmlenc11#aes128-gcm
· XML signatures according to  http://www.w3.org/2001/04/xmlenc#sha256
· MIKEY SAKKE/ECCSI (RFC 6507, 6508, 6509)
· LTE Band 1
· EEA2/EIA2 RRC/NAS ciphering/integrity algorithms
[bookmark: _Toc68002783][bookmark: _Toc93054718]Execution Log Files 
[bookmark: _Toc54888067][bookmark: _Toc68002784][bookmark: _Toc93054719]Nemergent Solutions Client v7.0
The Nemergent Solutions Client v7.0 passed this test case against the MCSTaaSting Tester. The documentation below is enclosed as evidence of the successful test case run [1]:
· SSLog_Coversheet.txt
· Signalling-TC5.4.MCPTT.html (RRC/NAS signalling, IP flow decrypted/decoded)
· MCX-PIXIT.cfg (Titan configuration and PIXITs)
· TLI Interface log

	R5s220130: Supporting information for the addition of MCPTT test case 5.4 to the MCX ATS





